get file 'a:testdata.sys'.

* The following run provides data for scoring checks.

freq variables = all.

* The following recodes all the reversed items.

recode move4 move5 move6 move7 move10 move16
    move17 move18 (0=1) (1=0).

recode home2 home4 (0=1) (1=0).

recode emot5 emot6 emot7 (0=1) (1=0).

* The following recodes all the missing codes to 0's without recoding the patient
  * self-rating items, JOBYN, or JOBHLTH.
* In this program, we have used the following missing codes
  * 7 = not applicable, true skip
  * 8 = missing
  * 9 = don't know

recode move1 to move20 fine1 to fine7 home1 to home9
    adl1 to adl18 sleep1 to sleep6 leis1 to leis4
    relat1 to relat10 cog1 to cog4 emot1 to emot18
    job1 to job4 (7,8,9, = 0).

* The following recodes JOBR based on responses to JOBYN and JOBHLTH.

if (jobyn = 0 and jobhlth =1) jobr = 5.
if (jobyn = 0 and jobhlth =0) jobr = 1.

* This recodes the patients self-rating items so that missing responses are coded with
  * missing value indicators.
* Because SPSSPC will only allow the use of one value for missing, this recode was
  * necessary. The new missing value for the patient self-rating items is 88.

recode moveR fineR homeR adlR sleepR leisR relatR cogR
    emotR jobr (7,8,9 = 88).

missing values moveR fineR homeR adlR sleepR leisR relatR cogR
    emotR jobR (88).

MFA Scoring Program for SPSS-PC - Page 2

* The following provides the default codes if item MOVE1 is "NO"(0)

if (move1 eq 1) move2 = 1.
if (move1 eq 1) move6 = 1.
if (move1 eq 1) move7 = 1.
if (move1 eq 1) move8 = 1.
if (move1 eq 1) move9 = 1.
if (move1 eq 1) move10 =1.
if (move1 eq 1) move12 =1.
if (move1 eq 1) move13 =1.
if (move1 eq 1) move18 =1.
* The following provides the default codes if item JOBHLTH is "YES" (1)

if (jobhlth eq 1) job1 = 1.
if (jobhlth eq 1) job2 = 1.
if (jobhlth eq 1) job3 = 1.
if (jobhlth eq 1) job4 = 1.

* The following are the raw scores for categories (MOWERAW TO JOBRAW) 
* and for all 100 items of the MFA (TOTRAW).

compute moveraw = move1 + move2 + move3 + move4 + move5 + move6 + 
move7 + move8 + move9 + move10 + move11 + 
move12 + move13 + move14 + move15 + move16 + 
move17 + move18 + move19 + move20.
compute fineraw = fine1 + fine2 + fine3 + fine4 + fine5 + fine6 + 
fine7.
compute homeraw = home1 + home2 + home3 + home4 + home5 + home6 + 
home7 + home8 + home9.
compute adlraw = adl1 + adl2 + adl3 + adl4 + adl5 + adl6 + 
adl7 + adl8 + adl9 + adl10 + adl11 + 
adl12 + adl13 + adl14 + adl15 + adl16 + 
adl17 + adl18.
compute sleepraw = sleep1 + sleep2 + sleep3 + sleep4 + 
sleep5 + sleep6.
compute leisraw = leis1 + leis2 + leis3 + leis4.

MFA Scoring Program for SPSS-PC - Page 3

*Continuing from Page 2
* The following are the raw scores for categories (MOWERAW TO JOBRAW) 
* and for all 100 items of the MFA (TOTRAW).

compute relatraw = relat1 + relat2 + relat3 + relat4 + relat5 + relat6 + 
relat7 + relat8 + relat9 + relat10.
compute cograw = cog1 + cog2 + cog3 + cog4.
compute emotraw = emot1 + emot2 + emot3 + emot4 + emot5 + emot6 + 
emot7 + emot8 + emot9 + emot10 + emot11 + 
emot12 + emot13 + emot14 + emot15 + emot16 + 
emot17 + emot18.
compute jobraw = job1 + job2 + job3 + job4.
compute totraw = moveraw + fineraw + homeraw + adlraw + sleepraw + 
leisraw + relatraw + cograw + emotraw + jobraw.

* The following are the standardized scores for categories (MOVESTD TO JOBSTD) 
* and for all 100 items of the MFA (MFASCORE).

compute movestd = (moveraw / 20)*100.
compute finestd = (fineraw / 7)*100.
compute homestd = (homeraw / 9)*100.
compute adlstd  = (adlraw  / 18)*100.
compute sleepstd =(sleepraw / 6)*100.
compute leisstd = (leisraw / 4)*100.
compute relatstd = (relatraw / 10)*100.
compute cogstd = (cograw / 4)*100.
compute emotstd = (emotraw / 18)*100.
compute jobstd = (jobraw / 4)*100.
compute mfascore = (totraw / 100)*100.

* The following provides the patient rating subscore (PRSCORE).
compute prscore = (moveR + fineR + homeR + adlR + sleepR +
                   leisR + relatR + cogR + emotR + JobR)/10.

MFA Scoring Program for SPSS-PC - Page 4

* The following provides labels for the above computed variables
variable labels moveraw 'raw score, mobility category'
   /fineraw 'raw score, hand-fine motor category'
   /homeraw 'raw score, housework category'
   /adlraw 'raw score, self-care category'
   /sleepraw 'raw score, sleep-rest category'
   /leisraw 'raw score, leisure, recreat category'
   /relatraw 'raw score, family relationships category'
   /cograw 'raw score, cognition-thinking category'
   /emotraw 'raw score, emotional adjustment category'
   /jobraw 'raw score, employment-work category'
   /totraw 'raw score, all 100 items from MFA'
   /movestd 'std score, mobility category'
   /finestd 'std score, hand-fine motor category'
   /homestd 'std score, housework category'
   /adlstd 'std score, self-care category'
   /sleepstd 'std score, sleep-rest category'
   /leisstd 'std score, leisure, recreat category'
   /relatstd 'std score, family relationships category'
   /cogstd 'std score, cognition-thinking category'
   /emotstd 'std score, emotional adjustment category'
   /jobstd 'std score, employment-work category'
   /mfascore 'std score, all 100 items from MFA'
   /prscore 'patient rating subscore'.

* This run provides information on the raw scores and standard scores.
* for the scoring checks
frequencies variables = moveraw to totraw movestd to prscore
   /statistics = all.

* EOF: SPSSPC.txt